

Exercícios XPath

Carlos A. Heuser

3 de Outubro de 2006

Para executar os exercícios abaixo, utilize o arquivo artigo.xml e a página XPathVsualizer (estão no Moodle). Construir expressões XPath para selecionar:

1. o elemento nome de cada autor do artigo

```
/artigo/autor/nome
```

2. os elementos filhos do elemento autor, que seguem ao nome do autor do artigo

```
/artigo/autor/nome/following-sibling::*
```

3. os elementos ancestrais do endereço do autor do artigo

```
/artigo/autor/endereco/ancestor::*
```

4. o texto (não o elemento) do nome do autor do artigo

```
/artigo/autor/nome/text()
```

5. os elementos nome de todos autores nas referências bibliográficas

```
/artigo/bibliografia/referencia/autor/nome
```

6. os textos de todos elementos autor dentro do artigo (tanto os autores do artigo, quanto os autores das referências)

```
//autor/nome/text()
```

7. todos elementos que seguem ao elemento resumo do artigo

/artigo/resumo/following::*

8. o atributo versão do artigo

/artigo/@versao

ou

/artigo/attribute::versao

9. os elementos autor que contém instituição

//autor/instituicao/..

ou

/artigo//autor[instituicao]

10. o elemento nome dos elementos autor que contém instituição

//autor/instituicao/../nome

ou

/artigo//autor[instituicao]/nome

11. os elementos seção cujo atributo título é “Results”

/artigo/secao[@titulo='Results']

ou

/artigo/secao[attribute::titulo='Results']

12. referência bibliográfica cujo id=b2

/artigo/bibliografia/referencia[@id='b2']

13. os elementos referência, cujo ano é 1999

/artigo/bibliografia/referencia[ano='1999']

ou

```
/artigo/bibliografia/referencia[ano=1999]
```

14. os nomes dos autores (texto, não elemento) das referências de 1999

```
/artigo/bibliografia/referencia[ano=1999]/autor/nome/text()
```

15. para cada obra de 1999, o elemento nome do primeiro autor da obra

```
/artigo/bibliografia/referencia[ano=1999]/autor[1]/nome
```

16. os elementos seção que contém mais que dois elementos parágrafo

```
/artigo/secao[count(paragrafo)>2]
```

17. o último elemento figura dentro de cada seção que contém mais que dois elementos parágrafo

```
/artigo/secao[count(paragrafo)>2]/figura[position()=last()]
```

18. os elementos parágrafo que contém '...'

```
/artigo/secao/paragrafo[contains(.,'...')]
```

ou

```
/artigo/secao/paragrafo[contains(text(), '...')]
```

19. os títulos (atributo) das seções que possuem parágrafos que contém 'XML'

```
/artigo/secao[paragrafo[contains(text(), 'XML')]]/@titulo
```

cuidado: a solução abaixo testa somente o primeiro parágrafo de cada seção, já que a função contains opera sobre um único nodo

```
/artigo/secao[contains(paragrafo, 'XML')]/@titulo
```

20. nomes dos autores que começam com 'Mary'

```
//autor[starts-with(nome, 'Mary')]/nome
```

ou

```
//autor/nome[starts-with(., 'Mary')]
```

21. referências em que ao menos um autor é o segundo autor da primeira obra

```
/artigo/bibliografia/referencia[autor/nome=  
/artigo/bibliografia/referencia[1]/autor[2]/nome]
```